ОБРАБОТКА     МАССИВОВ

СРЕДСТВАМИ ЯЗЫКА 

ПРОГРАММИРОВАНИЯ     ТР 7.0

1. Общие понятия
Существует множество задач, в которых большую группу данных нужно подвергнуть однотипной обработке.
Пример

В академической группе студентам, имеющим средний балл по результатам сессии выше 4, требуется увеличить размер сти​пендии на 25%.
В этом случае удобно всю группу средних оценок обозначить общим именем. Такую группу данных принято называть массивом чисел.
Массив — это переменная сложного структурированного типа данных, объединяющая под одним общим именем множество переменных более простого типа данных.
В предложенном выше примере назовём весь массив средних оценок именем X. Элементом массива является любое конкретное число, принадлежащее этому массиву. Средний балл каждого студента группы рассматривается как элемент массива средних оценок X.

Рис.1

Чтобы обработать какой-либо элемент этого массива, необходимо указать имя массива и номер элемента по порядку с начала массива. Этот номер называется индексом.
Пример
Х[3] — есть третий элемент массива X. Его значение равно средней оценке третьего студента по списку группы (т.е. 4.0).
В оперативной памяти компьютера элементы массива располагаются последовательно, занимая смежные ячейки памяти. Элементы с меньшим значением индекса хранятся в ячейках памяти с меньшим адресом. Количество памяти, отводимое под каждый элемент массива, определяется типом элементов массива (который, напомним, одинаков для каждого элемента массива). Таким образом, массив – это последовательность однотипных элементов, расположенных в смежных ячейках памяти.
Пример
Если обозначить значения средних оценок вещественными числами, 3.1, 4.5, 4.0 и т.д., то каждый элемент массива будет иметь тип real (т.е. занимать в памяти компьютера 6 байтов).
Количество элементов массива в Паскаль-программе должно быть строго фиксировано. Это связано с тем, что объём памяти, требуемый для массива, резервируется только один раз (в начале выполнения программы) и не может быть изменён до завершения программы.
2. Одномерные массивы (векторы)

2.1. Объявление одномерных массивов
Массивы, упорядоченные в памяти по одному порядковому номеру, называются одномерными массивами (векторами) (см. рис.1).
Т.к. одномерный массив является переменной, то в любой программе, использующей массивы, необходимо зарезервировать память для его размещения. Это действие выполняется обычным для всех переменных способом (т.е. в разделе описаний Паскаль-программы в подразделе var указывается имя массива и его тип):
var
X : аггау[<количество элементов>] of <базовый тип>;
Здесь <количество элементов> задаётся как диапазон возможных значений индексов. Например, 1..100— задаёт диапазон из ста индексов с указанными начальным и конечным значениями; 1901..2000 — тоже задаёт диапазон из ста индексов, но с другими границами.
<Базовый тип> может представлять собой любой оперативный тип данных.
Пример объявления одномерного массива.
Var

X: array [1. .50] of integer; (массив из 50 целых чисел) 

Y,Z:array [1..100] of real;{массивы из 100 вещественных чисел каждый}

2.2. Решение задач с использованием одномерных массивов
В общем случае алгоритм решения задачи с использованием массивов выглядит так, как показано на рисунке 2. Рассмотрим более подробно шаги этого алгоритма.

Рис.2
2.2.1. Ввод одномерного массива в память компьютера
При вводе массива предлагается, что память для его размещения уже зарезервирована. Количество элементов, под которые зарезервирована память, обозначим константой LEN. Тогда LEN — это максимально возможное для использования количество элементов массива (длина массива). В действительности при решении большинства задач количество требуемых элементов может изменяться от одного запуска программы к другому. Поэтому следует предоставить пользователю возможность заказать требуемое ему в данном конкретном случае число элементов массива, которое обозначим N. Таким образом, первым шагом алгоритма ввода одномерного массива станет ввод переменной N — требуемой длины массива. Напомним, что N всегда меньше либо равна LEN.
В языке Паскаль нет средств ввода-вывода всего массива целиком, поэтому ввод и вывод массивов выполняется поэлементно. Словесно можно описать шаг ввода одномерного массива следующим образом:
ввести элемент А[1];
ввести элемент А[2];

….

ввести элемент А[N].
Таким образом, индекс вводимого элемента изменяется от 1 до N с шагом 1. Очевидно, что для реализации ввода массива необходимо выбрать структуру цикла. Блок-схема ввода массива представлена на рисунке 3.

Рис.3
На языке ТурбоПаскаль ввод одномерного массива реализуется следующим фрагментом программы:
Program ABC; 

    const

        LEN = 100; {максимальная длина массива} 

    var

       А : array [1. .LEN]  of  real;  {объявление массива, состоящего из 100 вещественных чисел} 

       i,N : integer; {i — индекс, N — требуемая длина массива} 

   BEGIN

        write (‘Введите длину массива N ='); 

        read(N);

        for i:=1 to N do 

           begin

               write(‘A[‘ ,  i ,  ‘]='   );                    

               read (A [i]) 

           end;

  …

   END.

В дальнейшем заголовок и раздел описания программы приводить не станем. Фрагменты программ, описанные в примерах, следует располагать в требуемом месте раздела исполняемых операторов приведённой здесь программы.
2.2.2. Вывод массива на экран
Блок-схема алгоритма вывода одномерного массива на экран представлена на рисунке 4. Она очень мало отличается от вышеприведённой блок-схемы алгоритма ввода и построена на основании таких же рассуждений. Единственным принципиальным отличием является то, что не требуется запрашивать у пользователя длину массива, которая к моменту вывода массива на экран уже известна.
Фрагмент программы для вывода одномерного массива на экран:

writeln(‘Массив А’);

for i:=1 to N do         

   write (A[ i ]:6:2);

writeln;    

…

                   Рис.4

2.2.3. Типовые алгоритмы для обработки одномерных массивов
Т.к. числовые массивы являются наиболее распространённой структурой данных в инженерных задачах, было разработано большое количество типовых алгоритмов обработки массивов. В этом разделе будут рассмотрены некоторые из них.
2.2.3.1. Подсчёт суммы, произведения, количества элементов массива, удовлетворяющих заданному условию
Для нахождения суммы элементов массива воспользуемся методом накопления. Для этого заведём дополнительную переменную S, к которой поочерёдно станем добавлять значения элементов массива. Т.к. при объявлении любой переменной на языке ТурбоПаскаль её значение не определено, необходимо занести в переменную S такое значение, которое не повлияет на конечный результат суммирования элементов массива. Этим значением является 0.
Далее необходимо к значению переменной S добавлять элементы массива с номерами 1, 2, ..., N, т.е. выполнять следующие действия:

S = S +А[1]

S = S + A[2]

. . . . . . . . . . .

S = S + A[N]
Отсюда видно, что для вычисления суммы элементов массива необходимо использовать цикл, который будет повторять оператор S=S+A[i], где i поочерёдно принимает значения от 1 до N с шагом 1. После завершения N итераций цикла значение суммы будет вычислено окончательно и его можно обработать по условию задачи, например, вывести на экран.

Блок-схема алгоритма нахождения суммы приведена на рисунке 5. Фрагмент программы записан ниже:

Рис.5

В случае, если необходимо найти сумму элементов, удовлетворяющих некоторым условиям (положительные, отрицательные, четные и т.п.), то в этом случае каждый элемент массива необходимо будет сравнивать с некоторым значением, заданным в условии задачи. На рис.6 представлена блок-схема алгоритма поиска суммы положительных значений элементов массива. В этом случае каждый элемент массива сравнивается с 0, и, если он больше 0, то он включается в сумму. Фрагмент программы приводится ниже.

…                                                

S := 0;

for i :=1  to  N  do                        

   if  A[i] > 0  then

     S := S + A[i];

writeln (‘S=  ‘,S);

…









Рис. 6

На рис. 7 представлена блок-схема алгоритма поиска суммы четных элементов массива. Известно, что четные элементы делятся на 2 без остатка. В ТР существует математическая операция MOD, позволяющая определить остаток от деления числа а на некоторое число р: а MOD р. Следует заметить, что операнды а и р могут быть только целыми числами. Например: 

4 MOD 2 = 0 ( т.к. 4 – четное число);

                                5 MOD 2 = 1 ( т.к. 5 – нечетное число);

                                7 MOD 3 = 1 ( т.к. 7 не кратно 3 ).

Фрагмент программы приводится ниже.

…

S := 0;                                                       

for i :=1  to  N  do                        

   if  A[i] MOD 2 =  0  then

     S := S + A[i];

writeln (‘S=  ‘,S);

…

     Рис. 7

Аналогично следует вычислять произведение элементов массива, за исключением того, что в качестве начального значения произведения следует принять 1, т.к. именно умножение на 1 не повлияет на искомое произведение.

На рис. 8 представлена блок-схема алгоритма поиска произведения Р элементов массива, попадающих в интервал от –2 до 9. Фрагмент программы приводится ниже.

… 
P :=1;

for  i:=1 to N do

   if  (A[i] => -2)  and  (A[i] <= 9) then              

     P := P * A[i];

writeln(‘P=   ‘, P);

…

           Рис. 8

При нахождении количества элементов массива следует воспользоваться теми же рассуждениями, что и в ранее рассмотренных случаях. Обычно требуется подсчитать количество элементов, удовлетворяющих некоторым заданным условиям, например, количество отрицательных элементов. Для решения этой задачи поочерёдно сравним каждый элемент массива с 0, и при обнаружении отрицательного увеличим количество на 1. Когда все элементы массива закончатся, искомое количество будет окончательно подсчитано и его можно обработать по условию задачи.
Блок-схема алгоритма нахождения количества отрицательных элементов массива похожа на блок-схему на рисунке 6. Фрагмент программы, соответствующий этой блок-схеме, приведён ниже.
К:=0;

for i:=l to N do                  

if A[i]>0 then   K:=K+1;

writeln ( ‘Количество  равно’, К);

2.2.3.2. Поиск значений среднего арифметического и среднего геометрического элементов массива

Среднее арифметическое (обозначим как SA) всех значений элементов массива подсчитывается по формуле: 

SA = S / К,

где  S – это сумма всех элементов, K – количество просуммированных элементов. Алгоритмы поиска суммы и количества элементов были подробно рассмотрены в п/п 2.2.3.1. Но в тех случаях, когда SA находят только среди элементов, удовлетворяющих некоторому условию, следует учитывать тот факт, что таких элементов в массиве может и не быть, и тогда К = 0. Чтобы избежать деления на 0, необходимо при получении значения SA осуществлять соответствующую проверку: К > 0. На рис. 9 приведена блок-схема алгоритма поиска SA положительных  элементов и фрагмент программы.

Среднее геометрическое значений массива (обозначим через SG) вычисляется по следующей формуле:

SG = K( P,

где Р – произведение элементов массива, K – количество элементов. Аналогично, как и в алгоритме поиска SА элементов, удовлетворяющих некоторому условию, необходимо учитывать вариант К = 0. На рис. 10 приведена блок-схема алгоритма поиска среднего геометрического отрицательных элементов массива и соответствующий фрагмент программы.

S:= O;

K := 0;

for i:=1 to N do

   if a[i]> 0 then

     begin

        S := S + A[i];                           

        K := K +1;

    end;

if  K > 0 then

   begin

     SA := S/K;

     writeln (‘ SA=  ‘, SA);

   end

else

   writeln(‘ K = 0 ‘); 

…     *

       Рис. 9

P:= 1; 
K := 0;

for i:=1 to N do

   if A[i]< 0 then

     begin

         P := P* A[i];

         K := K +1;

     end;                                            

if  K > 0 then

  begin

      SG := EXP(1/K * LN(P)) ;

      writeln (‘ SG=  ‘, SG);

  end

else 

   writeln(‘ K = 0 ‘);

…

  






                    Рис. 10

2.2.3.3. Обработка элементов одномерного массива, имеющих чётные индексы
В некоторых задачах требуется обработать не все подряд элементы массива, а лишь те, индексы которых кратны определённому числу.
Пусть, например, требуется умножить на 2 значения элементов с чётными индексами. Понятно, что обработку массива придётся начать с элемента, имеющего номер 2, т.е. переменная i, обозначающая индекс элемента, примет начальное значение, равное 2. Выбирать из массива элементы для обработки следует «перешагивая» через элемент, т.е. индекс i будет изменяться с шагом 2.

Блок-схема описанного алгоритма приведена на рисунке 11. Фрагмент программы записан ниже. В нём использован не привычный уже цикл со счётчиком (цикл FOR), а универсальный цикл «пока» (цикл WHILE), который, в отличие от FOR, приспособлен к работе с любым шагом (в том числе и с шагом 2).

i:=2;

while  i < =N  do

    begin

       A[ i ] := A[ i ] * 2;

i:=i+2;

   end;

…

                                                              Рис.11

Следует обратить внимание, что в случае работы с нечетными индексами начальное значение i:=1, i будет изменяться с шагом 2. С помощью подобных алгоритмов можно обрабатывать элементы массива с индексами, кратными любому числу. Например, найти сумму элементов с индексами, кратными 3 ( i=3; шаг +3) заменить нулём каждый четвёртый элемент массива и т.п.

2.2.3.4. Формирование нового одномерного массива из элементов другого одномерного массива

Рассмотрим этот алгоритм на примере следующей задачи: требуется сформировать из отрицательных элементов массива А новый массив В.
Так как общее число элементов и количество отрицательных элементов массива А зависят от данных, вводимых пользователем в процессе выполнения программы, то число элементов в массиве В заранее неизвестно. Поэтому в процессе формирования массива В необходимо подсчитывать количество встречающихся в массиве А отрицательных элементов. Номер очередного записываемого в массив В элемента будет равен значению количества уже встреченных в массиве А отрицательных элементов. Общее количество элементов массива В будет равно количеству отрицательных элементов массива А. Этот факт следует учитывать при выводе массива В на экран (в условии цикла диапазон изменения переменной i станет равен 1..К).
Блок-схема алгоритма формирования массива В приведена на рисунке 12. Фрагмент программы для формирования массива В и его распечатки показан ниже.

{Формирование   массива В} 
К:=0;

for   i:=1  to  N  do 

    if  A[i] < 0 then                  

     begin

          K:=K+1; 

          B[K] := A[I];

      end;

{Вывод массива на печать}

writeln(‘Сформированный массив В’);

for i:=1 to K do

   writeln(‘B[‘,i,’]=  ‘,B[i]);

…

  





           Рис.12


2.2.3.5. Нахождение максимального (минимального) элемента одномерного массива
Принцип нахождения максимального и минимального элементов массива одинаков. Поэтому рассмотрим его на примере алгоритма поиска максимального элемента.
Нахождение максимального элемента массива выполняется путём последовательных сравнений. Значение наибольшего из сравниваемых элементов станем запоминать в переменной МАХ. В качестве начального значения максимума примем значение первого элемента массива. Значение переменной МАХ будем сравнивать с каждым следующим элементом массива. Если МАХ окажется меньше очередного элемента, то значение этого элемента сохраним в переменной МАХ для сравнения с последующими элементами массива. В противном случае значение переменной МАХ оставим без изменений.

Таким образом, когда все элементы массива будут просмотрены, в переменной МАХ останется копия максимального элемента массива. Сам же массив изменён не будет.
Блок-схема алгоритма нахождения максимума приведена на рисунке 13. Фрагмент программы приведён ниже.

MAX := A[1];

for   i:=2   to  N  do 

   if  MAX<A[i]    then  

      MAX:=A[i]; 

writeln('МАХ=',   MAX);










    Рис.13

Ещё раз напомним, что с помощью вышеописанного алгоритма будет сформирована копия значения максимального элемента одномерного массива.
При этом ничего не будет известно о местоположении максимального элемента в массиве, т.е. об его индексе. Между тем существует большая группа задач, для решения которых необходимо знать номер максимального элемента в массиве. В этот класс задач входят те, в которых требуется преобразовать сам максимальный элемент массива; задачи, в которых требуется обработать часть массива до максимального элемента или после него и т.п.
Алгоритм нахождения номера максимального элемента в одномерном массиве показан на рисунке 14. Понятно, что номер максимального элемента следует находить одновременно со значением максимума.
 
Суть алгоритма нахождения номера максимального элемента массива состоит в том, что при изменении значения, хранящегося в переменной МАХ, на значение большего элемента массива ещё в одной переменной Imax запоминается индекс элемента, принятого за максимальный. После просмотра всех элементов массива в переменной МАХ останется копия значения максимального элемента массива, а в переменной Imax – номер максимального элемента. Чтобы преобразовать максимальный элемент массива, необходимо обратиться к нему по его найденному номеру. В качестве примера на рисунке 14 приведён алгоритм нахождения номера максимального элемента одномерного массива и замена максимального элемента нулём. Фрагмент программы приведён ниже.

МАХ:=А[1] ;

Imax:=1;

for   i:=2   to  N  do 

   if  MAX<A[i]    then     

     begin

MAX:=A[i];

Imax:=i; 

     end;

A[Imax]:=0;

…







             Рис.14

Как указывалось выше, алгоритм поиска минимального значения

MIN массива похож на алгоритм поиска максимального значения. Аналогично, в качестве начального значения MIN можно принять значение первого элемента массива, а затем MIN поэлементно сравнивать с элементами массива. Если значение MIN окажется больше элемента массива, то значение этого элемента массива необходимо сохранить в переменной MIN. Фрагмент программы поиска минимального значения среди всех элементов массива приводится ниже.

…

MIN := A[1];

for  i := 2  to  N do

   if MIN > A[i]  then

     MIN := A[i];

writeln (‘MIN=  ‘, MIN);

…


В случае, когда требуется найти MIN и MAX не среди всех элементов массива, а лишь среди тех элементов, которые удовлетворяют некоторому условию (положительные, четные и т.п.), тогда в качестве начальных значений MIN и MAX нельзя принимать значение первого элемента массива, так как заранее не известно, удовлетворяют ли они заданным условиям. В этом случае необходимо поступить следующим образом: в качестве начального значения MAX принять заведомо малое число, которое не будет принадлежать данному массиву (например, MAX = -10000, и если необходимо, Imax = 0), а в качестве начального значения MIN – большое число (например, MIN = 10000, Imin = 0). Такой метод выбора значений MAX, MIN является универсальным.

2.2.3.6. Изменение порядка следования элементов в массиве на обратный (инверсия)


Пусть нам дан массив А = /1, 2, 3, 4, 5, 6/. В результате выполнения операции инверсии получим массив А = /6, 5, 4, 3, 2, 1/. Обмен элементов необходимо производить в цикле FOR, конечное значение параметра этого цикла равно количеству пар элементов, подлежащих обмену. Действительно, в нашем случае для достижения требуемого результата необходимо переставить местами A[1] и A[6], A[2] и A[5], A[3] и A[4] элементы, что соответствует К = 3 парам. Подсчитать количество пар, между которыми необходимо произвести обмен можно следующим образом: K= N DIV 2, где N – количество элементов массива, а DIV – математическая операция целочисленного деления (нельзя использовать вещественное деление /, так как параметр цикла FOR К должен быть целым числом). На рис. 15 приведена схема обмена между элементами произвольной пары A[1] и A[6].

                                       A[1]                                     А[6]

Рис. 15

Переменная V позволяет сохранить значения элемента массива A[1] при копировании в эту ячейку значения A[6]. Аналогично происходит обмен между другими парами. Формула обмена элементов выглядит следующим образом: A[i]           A[N+1-i]. Действительно, при N = 6 имеем:

i = 1 обмен между A[1] и A [6] ( 6+1-1= 6);

i = 2 обмен между A[2] и A [5] ( 6+1-2= 5);

i = 3 обмен между A[3] и A [4] ( 6+1-3= 4).

[image: image30.png]


Блок-схема алгоритма инверсии массива приведена на рис. 16. Фрагмент программы приводится ниже.

… 
K := N  DIV 2;

for i :=1 to K  do

   begin

      V := A[i];                       

      A[i] := A[N+1-i];

      A[N+1-i]  := V;

  end;

…                                                                                                Рис. 16

2.2.3.7. Сортировка элементов массива

Под сортировкой массива понимают процесс перестановки элементов массива определенным образом. Для числовых массивов это означает упорядочивание массивов либо по возрастанию, либо по убыванию. Существуют различные метода сортировки массивов. Рассмотрим наиболее распространенные: метод линейного перебора и метод «пузырька».


Суть метода линейного перебора заключается в следующем:

1. В исходном массиве выбирается элемент с наименьшим (если сортировка по возрастанию), либо с наибольшим значением (если сортировка по убыванию).

2. Этот элемент меняется местами с первым элементом массива. Он занимает свою окончательную позицию и в дальнейшем исключается из рассмотрения. Теперь исходный массив укорачивается слева на 1 элемент.

3. Шаги 1-2 повторяются до тех пор, пока длина укороченного массива больше одного элемента.

Очередной просмотр массива в соответствии с указанным методом называется проходом.

На рис. 17 приведен тестовый пример сортировки числового массива по убыванию.

Исходный массив           N = 5

Первый проход

Четвертый  проход

                     Результирующий  массив 

Рис. 17

На рис. 17 темным цветом отмечена ячейка, с которой начинается поиск максимального элемента в каждом проходе. В каждом проходе приводится массив после обмена элементов. Как видно из тестового примера, для упорядочивания массива необходимо N-1 проходов просмотра массива, причем поиск максимального значения в укороченном массиве начинается с того элемента, номер которого совпадает с номером прохода. В приведенном тестовом примере:

Первый проход

Перебирают массив с первого элемента. Максимальное значение массива равно A[3] = 15, поэтому производят обмен A[1] = -2 и A[3] = 15.

Второй проход

Перебирают массив со второго элемента. Максимальное значение массива равно A[2] = 8, поэтому не производят обмен.

Третий проход

Перебирают массив с третьего элемента. Максимальное значение массива равно A[6] = 6, поэтому производят обмен A[3] = -2 и A[6] = 6.

Четвертый проход

Перебирают массив с четвертого элемента. Максимальное значение массива равно A[6] = -2, поэтому производят обмен A[4] = -4 и A[6] = -2.

На рис. 18 приведена блок-схема алгоритма упорядочивания по убыванию методом линейного перебора. Внешний цикл по параметру J предназначен для счетчика проходов, а внутренний по I – для просмотра элементов массива внутри каждого прохода. Перестановка пар элементов аналогична процедуре, представленной на рис. 15. Фрагмент программы приводится ниже.

…

for j:=1 to N-1 do

   begin

       MAX := A[j];

       Imax := j;

       for i := j+1 to N do

          if A[i] > MAX  then

          begin

              MAX:= A[i];

              Imax:=i;                    

          end;  

       V:= A[j];

       A[j]:= A[Imax];

       A[Imax]:=V;

end;

…

Рис.18

      Если массив необходимо упорядочить по возрастанию, то необходимо в каждом проходе находить MIN.


Упорядочить массив методом «пузырька»  возможно также за N-1 проходов, только в каждом проходе массив просматривается полностью. Переставляют соседние элементы по следующему правилу:

1. Если массив упорядочивают по убыванию, то в паре соседних элементов на первое место ставится больший по значению элемент.

2.  Если массив упорядочивают по возрастанию, то в паре соседних элементов на первое место ставится меньший по значению элемент.

На рис. 19 приведен тестовый пример упорядочивания массива методом «пузырька»  по возрастанию.

Исходный  массив           N = 4

Третий проход    J = 3

Результирующий массив

Рис. 19

В каждом проходе приводится массив после процедуры обмена элементов.

На рис. 20 приведена блок-схема алгоритма упорядочивания по возрастанию методом «пузырька». Внешний цикл по параметру J предназначен для счетчика проходов, а внутренний по I – для просмотра элементов массива внутри каждого прохода (конечное значение параметра цикла I = N-1 – это максимальное количество пар обмена). Перестановка пар элементов аналогична процедуре, представленной на рис. 15. Фрагмент программы приводится ниже.

Рис. 20

3. Двумерные массивы (матрицы) 

3.1. Объявление двумерных массивов

По определению, элементом массива может быть переменная любого оперативного типа данных. Следовательно, можно объявить одномерный массив, состоящий из одномерных массивов, что в итоге даст таблицу или двумерный массив (рис. 21). 

X

	5


Рис.21

Чтобы обработать элемент матрицы, необходимо указать имя матрицы, а также номера строки и столбца, на пересечении которых он находится. Например, Х[2,3] — это элемент матрицы X, расположенный на пересечении 2 строки и 3 столбца. Таким образом, первый индекс всегда указывает номер строки, второй — всегда номер столбца.
Для объявления двумерного массива, состоящего из элементов вещественного типа, в разделе var Паскаль программы необходимо записать оператор: 

var

          X: array [1. .3, 1. .4]   of   real;

3.2. Решение задач с использованием двумерных массивов

Общий вид алгоритма решения задач такого рода задач смотрите на рис.2.

3.2.1. Ввод и вывод на экран двумерного массива
При вводе двумерного массива, также как и одномерного, необходимо разместить его в памяти. При этом необходимо указать размерность массива, исходя из максимально возможного количества строк и столбцов. Перед заполнением массива данными пользователь должен указать действительно необходимые ему количества строк и столбцов, которые обозначим соответственно N и М. Затем начинается непосредственный ввод элементов матрицы. Т.к. матрица состоит из N одномерных массивов (строк), то ввод одной строки следует повторить N раз. Тогда i — номер очередной обрабатываемой строки. Ввод строки с номером i аналогичен вводу элементов одномерного массива. Номер очередного элемента в строке (номер столбца) обозначим переменной j. Тогда очередной обрабатываемый элемент массива, расположенный на пересечении i-й строки и j-ro столбца, обозначается X[i,j]. Блок-схема алгоритма ввода матрицы приведена на рис. 22.


[image: image4.wmf]J=1,M

I=1,N

ВВОД

X[I,J]

ВВОД

N,M


Рис.22

В приведённом алгоритме элементы матрицы обрабатываются по строкам, т.е. переход к следующей строке происходит тогда, когда закончена обработка элементов предыдущей строки.
Для построения алгоритма вывода матрицы воспользуемся аналогичными рассуждениями. Блок-схема алгоритма вывода матрицы приведена на рис. 23.

Рис.23

Фрагмент программы для ввода и вывода матрицы приведён ниже.
Program BC; 

const

ROW  =  20; {Максимальное количество строк}

COL  =   30; {Максимальное количество столбцов}

var

X:array[1..ROW,1..COL]   of   real; 

N,M: integer; {требуемые количества строк и столбцов}

i, j : integer; {индексы строк и столбцов}

 BEGIN
      write ( ‘Введите количество строк:') ; 

      read (M) ; 

      write (‘Введите количество столбцов:') ; 

      read (N) ; 

      for  i:=1   to  N do 

         for   j:=1   to  M  do 

            begin

        write ('X[' , i , ' , ' , j , ']='); 

        read(X[i,j]); 

     end; 

         writeIn ( ‘Ввод матрицы завершен') ;

         {Вывод матрицы} 

         writeln(' Матрица X:');

          for   i:=1   to  N  do       

             begin

         for   j:=1   to  M do

         write(X [i,j] :6 :2) ; 

     writeln; {перевод курсора в следующую строку экрана после завершения вывода очередной cтроки матрицы}

 end;

        writeln (‘Вывод матрицы завершён’) ;
  END.
3.2.2. Нахождение суммы, произведения элементов, максимума всей матрицы в целом
В задачах подобного вида результатом является единственное значение. Поэтому начальные значения вычисляемым переменным присваиваются один раз — в начале алгоритма, до организации циклов обработки элементов массива. На рис. 24 показан алгоритм вычисления суммы элементов матрицы. Начальное значение сумме S=0 присваивается до начала суммирования элементов. Затем обработка элементов массива идёт по тому же принципу, что и в предыдущем пункте. 

S:=0;

for  i:=1 to N do 

   for j:=1 to M do

       S := S+X[i,j];

writeln(‘S=  ‘,S);

…









  Рис. 24

[image: image51.wmf]I=2,N

A[I]>MAX

+

-

MAX=A[1]

MAX=A[I]

ВЫВОД

MAX

Также рассуждаем и при нахождении максимального элемента матрицы (рис. 25). В качестве начального значения максимума принимаем её первый элемент, в качестве начальных значений координат максимума — единицы. Затем поочерёдно сравниваем максимум со всеми элементами матрицы построчно и запоминаем координаты вновь принятого за максимум элемента. При завершении обоих циклов максимум считается окончательно найденным и может быть обработан по условию задачи. Ниже приведены фрагменты программ, иллюстрирующие рис. 25.

MAX := X[1,1];

Imax:=1;

Jmax:=1;

For I:=1 to N do

    For j:=1 to M do

        If MAX < X[I,j] then

           Begin

                MAX:= X[I,j];

                Imax :=I;

                Jmax := j;

            End;

Writeln(‘max=  ‘,MAX);

…

 







     Рис. 25

 
3.2.3. Нахождение суммы, произведения, максимума в каждой строке матрицы
При решении задач подобного типа в ответе получается столько же значений, сколько в матрице строк. Поэтому начальные значения вычисляемым переменным присваиваем при каждом переходе к обработке следующей строки, т.е. между циклами. Результат для каждой строки также окончательно вычислен при завершении каждой строки и может быть обработан по условию задачи между циклами (после завершения цикла j).

На рис. 26 представлен алгоритм нахождения суммы элементов каждой строки матрицы X и вывод этих сумм на экран. Ниже приведены фрагменты программ, иллюстрирующие этот алгоритм.

… 
for i:=1 to N do

    begin

        S:=0;

           for  j: =1  to  M  do                            

               S:=S+X[ i , j ];

          writeln(‘S=  ‘ ,S);

      end;














   Рис. 26

На рис. 27 приведён алгоритм вычисления максимума каждой строки матрицы X и замена его нулём. 

for i:=l to N do 

   begin 

        MAX:=X[i,1] ;        Jmax:=1;

       for j:=2 to M do 

           if  X[I,j] > MAX then  

  begin

MAX:=X[i,j];Jmax:=j             end;          X[i,Jmax] :=0; 

      end;









   Рис.27

3.2.4. Нахождение суммы, произведения, максимума в каждом столбце матрицы
3.2.4. Алгоритм обработки элементов каждого столбца

При решении задач подобного типа в ответе получается столько же значений, сколько в матрице столбцов. В качестве примера на рисунке 28 приведён алгоритм увеличения в 2 раза максимальных элементов столбцов матрицы X. Комментарии к этому алгоритму аналогичны комментариям к предыдущему пункту. Ниже приведён фрагмент программы к этому алгоритму.

for j:=1 to M do 

begin
    МАХ:=Х[1,j];

    Imax:=1;

    for i:=2 to N do       if  MAX<X[i,j]    then                 

         begin

             MAX:=X[i,j]; 

             Imax:= i; 

         end;

     X [Imax, j ]: =X [Imax, j ] *2; 

 end;

…








Рис.28

3.2.5. Обработка элементов главной диагонали
Частным случаем матрицы является квадратная матрица, т.е. матрица, имеющая одинаковое количество строк и столбцов (рис. 29). 

	
	
	

	
	
	

	
	
	


Рис.29

Элементы матрицы, выделенные на рисунке, называются главной диагональю. Отличительной особенностью этих элементов, является то, что они имеют одинаковые номера строки и столбца, которые можно обозначить одной переменной. Тогда общий вид элемента главной диагонали такой: X[i,i]. Для обработки элементов главной диагонали требуется изменять только одну переменную, поэтому необходимо всего один цикл, а не два, как при обработке всей матрицы. На рисунке 30 показан алгоритм нахождения суммы положительных элементов главной диагонали. Ниже приведён фрагмент программы к [image: image55.wmf]K=N DIV 2

этому алгоритму.

•    •    • 
S:=0;

for   i:=1   to  N do 

   if  X[i,i]>0   then                            

      S:=S+X[i,i];                     

writeln(‘S='    ,   S);                              

                                                                                     Рис.30

•    •    •


3.2.6. Обработка элементов побочной диагонали 

Побочная диагональ (рис. 32) существует только в квадратной матрице. 

	
	
	

	
	
	

	
	
	


Рис.32

Для обработки элементов побочной диагонали удобен обход, показанный на рисунке 31. В этом случае номера строк увеличиваются от 1 до N, а номера столбцов уменьшаются от N до 1. При этом строки и столбцы заканчиваются одновременно, следовательно, требуется контролировать ситуацию завершения либо только строк, либо только столбцов.
На рисунке 31 показан алгоритм нахождения количества равных нулю элементов побочной диагонали. Ниже представлен фрагмент программы к этому алгоритму.

•    •    • 
К:=0; 

j : =N;

for   i:=1   to  N do     

   begin

       if  X[i,j] = 0   then

          K:=K+1; 

       j:=j-1; 

   end; 

writeln( ‘K='    ,   K) ;

•    •    • 







Рис.31

3.2.7. Обработка элементов квадратных матриц, расположенных над и под главными и побочными диагоналями 

Как было указано ранее, во многих задачах требуется обрабатывать не все элементы матрицы, а лишь определенные из них. Существует целый класс задач, в которых работают с элементами, расположенными над и под главной и побочной диагоналями квадратной матрицы. Особенностью этих алгоритмов является то, что необходимо определенным образом задавать начальные и конечные значения параметров цикла.

На рисунке 33 приведена схема расположения элементов относительно побочной диагонали, а ниже – пример задания циклов (элементы побочной диагонали не учитываются).

                                 Элементы, расположенные под ПД

Рис. 33

{Обработка элементов над ПД}

FOR I :=1 TO N-1 DO

     FOR J := 1 TO N-I DO



……………….

{Обработка элементов под ПД}
FOR I :=2 TO N DO

    FOR J := N+2-I TO N DO



……………….

На рисунке 34 приведена схема расположения элементов относительно главной диагонали, а ниже – пример задания циклов (элементы главной диагонали не учитываются).

Элементы, расположенные под ГД        Рис. 34

{Обработка элементов над ГД}

FOR I :=1 TO N-1 DO

     FOR J := I+1 TO N DO 



……………….

{Обработка элементов под ГД}
FOR I :=2 TO N DO

     FOR J := 1 TO I-1 DO



……………….

3. Пример решения задач

Задача 1

Дан одномерный массив T(N). Найти в массиве минимальный элемент и сумму четных элементов. Затем заменить этой суммой все отрицательные элементы, стоящие после минимального элемента.

Тестовый пример

Ввод:

Размерность массива N=10.

Массив Т: {13; 15; 11; –100; 8; 4; –7; 6; –5; 78}

Вывод:
Минимальный элемент массива: –100.

Сумма четных элементов массива: –4.

Массив Т после преобразования: {13; 15; 11; –100; 8; 4; –4; 6; –4; 78}

Блок-схема с пошаговой детализацией

Программа на языке высокого уровня ТР7.0
PROGRAM ZADACHA1;

  CONST

       N1=50;

  TYPE

       MAS1=ARRAY[1..N1] OF INTEGER;

  VAR

       T:MAS1;

       IMIN,MIN,S,I,N:INTEGER;

  BEGIN

{ШАГ 1-2}

       WRITE('ВВЕДИТЕ РАЗМЕР N=');

        READ(N);

        FOR I:=1 TO N DO

             BEGIN

                   WRITE('T[',I,']= ');

                   READ(T[I]);

             END;

{--------------------------------------------------------}

       {ШАГ 2-3}

        MIN:=T[1];

        IMIN:=1;

        FOR I:=2 TO N DO

             IF T[I]<MIN THEN

                BEGIN

                     MIN:=T[I];

                     IMIN:=I;

                END;

        WRITELN('МИНИМАЛЬНЫЙ ЭЛЕМЕНТ МАССИВА: ',MIN);

    {--------------------------------------------------------}

        {ШАГ 3-4}

        S:=0;

        FOR I:=1 TO N DO

             IF T[I] mod 2=0 THEN

                S:=S+T[I];

        WRITELN('СУММА ЧЕТНЫХ ЭЛЕМЕНТОВ МАССИВА: ',S);

     {--------------------------------------------------------}

        IF IMIN<N THEN

           BEGIN

     {ШАГ 4-5}

                FOR I:=IMIN+1 TO N DO

                     IF T[I]<0 THEN

                        T[I]:=S;

   {--------------------------------------------------------}

     {ШАГ 5-6}

        FOR I:=1 TO N DO

             WRITELN('T[',I,']=',T[I]:4);

          END
  {--------------------------------------------------------}

      ELSE
           WRITE('МИНИМАЛЬНЫЙ ЭЛЕМЕНТ ПОСЛЕДНИЙ');

      READLN;

END.
Задача 2

В массиве В(N) найти максимальный элемент. Если максимальный элемент находится в первой половине массива, то сформировать массив К, переписав в него все отрицательные элементы массива В, и массив Р, переписав в него все положительные элементы массива В. Массивы К и Р отпечатать.

Тестовый пример

Ввод:

Размерность массива N=15.

Массив Т: {13; -15; -11; 100; 8; 4; 0; 6; –5; 0; 3; -8; 4; -3; 1}

Вывод:

Номер максимального элемента массива В: 4.

Массив К: {-15; -11; -5; -8; –3} Массив К: {13; 100; 8; 4; 6; 3; 4; 1}

Блок-схема с пошаговой детализацией

Программа на языке высокого уровня ТР7.0
PROGRAM ZADACHA2;

     CONST

           N1=50;

     TYPE

        MAS1=ARRAY[1..N1] OF REAL;

   VAR

         B,K,P:MAS1;

         MAX:REAL;

         IMAX,K1,K2,I,N:INTEGER;

   BEGIN

       {ШАГ 1-2}

        WRITE('ВВЕДИТЕ РАЗМЕР N=');

        READ(N);

        FOR I:=1 TO N DO

             BEGIN

                  WRITE('B[',I,']= ');

                  READ(B[I]);

             END;

{-----------------------------------------------------------------------} 

      {ШАГ 2-3}

       MAX:=B[1];

       IMAX:=1;

       FOR I:=2 TO N DO

            IF B[I]>MAX THEN

              BEGIN

                   MAX:=B[I];

                   IMAX:=I;

             END;

{-----------------------------------------------------------------------}

      IF IMAX<(N div 2)+1 THEN

         BEGIN

     {ШАГ 3-4}

              K1:=0;

              K2:=0;

              FOR I:=1 TO N DO

                   IF B[I]<0 THEN

                      BEGIN

                           K1:=K1+1;

                           K[K1]:=B[I];

                      END

                   ELSE

                        IF B[I]>0 THEN

                           BEGIN

                                K2:=K2+1;

                                P[K2]:=B[I];

                           END;

{-----------------------------------------------------------------------}

      {ШАГ 4-5}

      FOR I:=1 TO K1 DO

           WRITELN('K[',I,']=',K[I]:6:2);

{-----------------------------------------------------------------------}

{ШАГ 5-6}

     FOR I:=1 TO K2 DO

           WRITELN('P[',I,']=',P[I]:6:2);

{-----------------------------------------------------------------------}

  END;

END.

Задача 3

Дана матрица K(N,M). Найти среднее арифметическое в каждой строке матрицы и переписать в одномерный массив Т. Затем в массиве изменить порядок следования элементов на обратный.

Тестовый пример

Ввод:

Размерность массива N=4, М=3.

Матрица К:
13 –15   8




  6     0   3




 -1    8    5




11   13   6

Вывод:

Массив Т: {2; 3; 4; 10}

Массив Т после перестановки: {10; 4; 3; 2}

Программа на языке высокого уровня ТР7.0

PROGRAM ZADACHA3;

CONST

     N1=20;

     M1=25;

TYPE

     MAS2=ARRAY[1..N1,1..M1] OF REAL;

     MAS1=ARRAY[1..N1] OF REAL;

VAR

       K:MAS2;

       T:MAS1;

     C,S:REAL;

     I,J:INTEGER;

     N,M:INTEGER;

BEGIN

{ШАГ 1-2}

     WRITE('ВВЕДИТЕ КОЛ-ВО СТОРОК N=');

     READ(N);

     WRITE('ВВЕДИТЕ КОЛ-ВО СТОЛБЦОВ M=');

     READ(M);

     FOR I:=1 TO N DO

         FOR J:=1 TO M DO

             BEGIN

                  WRITE('K[',I,',',J,']= ');

                  READ(K[I,J]);

             END;

{------------------------------------------}

{ШАГ 2-3}

     FOR I:=1 TO N DO

         BEGIN

              S:=0;

              FOR J:=1 TO M DO

                  S:=S+K[I,J];

              T[I]:=S/M;

         END;

{------------------------------------------}

{ШАГ 3-4}

     FOR I:=1 TO N DO

     WRITELN('T[',I,']=',T[I]:6:2);

{------------------------------------------}

{ШАГ 4-5}

     FOR I:=1 TO N DIV 2 DO

         BEGIN

              C:=T[I];

              T[I]:=T[N+1-I];

              T[N+1-I]:=C;

         END;

{------------------------------------------}

{ШАГ 6-7}

     FOR I:=1 TO N DO

          WRITELN('T[',I,']=',T[I]:6:2);

END.

